
0 メタな質問

0.1 このドキュメントはどこで手に入れられますか

このドキュメントの最新版は、Web上の http://www.gnuplot.info/faq/にあります。

0.2 このドキュメントに関するコメントはどこに送れば良いですか

コメント、提案等は開発者向けメーリングリスト mailto://gnuplot-beta@lists.sourceforge.netに送ってくだ
さい。

1 一般的な質問

1.1 gnuplotとは何ですか

gnuplotはコマンド駆動型の描画プログラムです。関数の描画、およびデータ点の描画を、2次元、3次元
の両方で、そして多くの異なるスタイル、多くの異なる出力形式で行うことができます。また、gnuplot
は自動的にグラフを生成するスクリプト言語としても使用することができます。これは主に科学的なデー
タを画面表示するものとして設計されています。gnuplotの著作権は保護されていますが、自由に配布
でき、代金を払う必要はありません。ソースコードのダウンロードも歓迎します。

1.2 それはどこから来て、そしてなぜ gnuplotと呼ばれているのですか

gnuplotの作者は Thomas Williams, Colin Kelley, Russell Lang, Dave Kotz, John Campbell, Gershon Elber,
Alexander Woo、そして多くの協力者です。

以下は Thomas Williamsによります:

私は微分方程式の講義を、そして Colinは電磁気学の講義を取っていて、2人ともそれらに関
する数学が視覚化できればなあと考えていました。私達はある EE (Engineering Electronics)
VLSI研のシステム管理者として働いていて、そして、グラフィック端末とコーディングを行
なう時間がありました。その投稿は我々の期待以上に良く受け入れられ、そして、不完全で
したが、ファイルデータに関するいくつかのサポートに我々を駆り立てました。

GNUplot という紹介はどんなものも正しくはありません。このプログラムの本当の名前は
"gnuplot"です。ずいぶんと "Gnuplot"と書かれているのを見るでしょうが、それは我々の大
半が、固有名詞やタイトルであっても、文章を小文字で開始することに嫌悪を持っているか
らです。gnuplotは GNUプロジェクトや FSFとは、極わずかの意味を除けば少しも関係あ
りません。我々のソフトウェアは完全に独立にデザインされたもので、そして "gnuplot"と
いう名前は実は妥協によるものです。私は "llamaplot"と呼びたかったし、Colinは "nplot"と
呼びたかったのです。そして、我々は "newplot"がいいだろうということで合意しましたが、
しかしそのときその名前を持つ、絶対的におぞましい Pascalのプログラムを計算機科学部で
たまに使っていることを知りました。それで私は "gnuplot"がいいごろ合わせになるとして、
後で Colinもそれに同意したのです。

1.3 gnuplotは何を提供してくれますか

• 2次元での関数やデータの、点、折れ線、誤差表示、塗り潰した図形、ラベル文字列、ベクトルな
どの沢山の異なる要素を組み合わせた描画

• 極座標軸、対数軸、より一般の非線形軸の割り当てや媒介変数座標

• 温度分布、ビースウォームグラフ、バイオリングラフ、ヒストグラムなどのデータ表現

• 3次元でのデータの、点、折れ線、曲面の、たくさんのスタイル (等高線描画、網かけ)による描画

• 整数、実数、複素数での代数演算

• Marquardt-Levenberg最適法を用いたデータ駆動型の曲線回帰

• 多くのオペレーティングシステム、多くのグラフィックファイル形式出力、そして多くの出力装置
のサポート

• 広範囲にわたるオンラインヘルプ

• ラベル、タイトル、軸、データ点への TEXライクな書式による見出し付け

• 対話型入力形式のエディット機能とヒストリ (履歴)機能のサポート

1.4 gnuplotはスクリプトに適していますか

ええ。gnuplot は対話型実行中に追加コマンドを含むファイルを読み込むことができますし、既に存在
するファイルや標準入力からのコマンド列をパイプを使ってバッチモードでそれを処理することもでき
ます。gnuplotは、Octaveのような高水準な数学パッケージの背後で実行するグラフィックドライバとし
て使われていますし、cgiスクリプトでラップすることで容易にWeb起動型の描画生成ツールとして使
うこともできます。gnuplot は、なじみのある命令 if else continue break while for を用いた文脈駆動型、
データ駆動型の処理制御、および繰り返しをサポートしています。

1.5 gnuplotは私のコンピュータで動かすことができますか

gnuplotは、MS Windows, linux, unix, OSXなどの多くのプラットホーム上で、広く利用できます。現在
のソースコードは、VMS, Ultrix, OS/2, MS-DOSなどの古いシステムのサポートも保持していますが、16
ビットプラットホームはもうサポートしていません。

gnuplotのソースは、それなりに標準的 (ANSI/ISO C, POSIX等)な環境から多少外れたものの上でもコ
ンパイルできるでしょう。

1.6 ライセンスについて

gnuplotはボランティアの集団によって作られたものです。彼らは gnuplot、およびその使用の許可、不
許可に関する法的な文書を作成できません。また何の保証もありません。自分自身の責任で使用してく
ださい。

gnuplotは、無償であると言う意味でフリーウェアです。gnuplotをあなたが望むように使用し、改変す
ることができますが、改変したものを配布することについてはある種の制限がつきます。Copyrightとい
うファイルの modification (改変)と redistribution (再配布)の項目を読んでそれに従ってください。

1.7 gnuplotは FSFや GNUプロジェクトと何か関係があるのですか

gnuplotは FSFが作ったのでも FSFがメンテナンスしているのでもありません。一時期 FSFから配布
されていたこともありますが、しかし現在はこれは正しくありません。gnuplotは、GNU General Public
License (GPL)に全く保護されていません。

1.8 更なる情報はどこで手に入りますか

gnuplot Webページ http://www.gnuplot.infoを見てください。

英語以外の他の言語での文書や入門書もあります。最新のリストについては、http://gnuplot.sourceforge.

net/help.htmlの "Localized learning pages about gnuplot"の節をご覧ください。

2 インストール

2.1 gnuplotの最新バージョンは

gnuplotの最新の安定版は、2017年 9月に最初にリリースされた 5.2です。更新バージョン (パッチレベ
ル 1, 2, ...) は、だいたい 6ヶ月毎にリリースされます。gnuplotの開発版は、現在は 5.3となっています。

2.2 gnuplotはどこで手に入れられますか

初めに最も適切な場所は http://www.gnuplot.infoです。ここから、例えば SourceForge上の開発プロジェ
クトサイト http://sourceforge.net/projects/gnuplotなどの、色々なサイトの場所を見つけられるでしょう。

ソース配布物 ("gnuplot-5.2.6.tar.gz"または同様の名前)は公式の配布サイト http://sourceforge.net/projects/

gnuplotにあります。

2.3 gnuplotの最新の開発版はどこで手に入れられますか

gnuplotの開発版は、gitリポジトリに保持されていて、以下に示すようにクローンを手元で作成し、ソー
スを調べたり、実行プログラムをソースから作成したりできます。

git clone https://git.code.sf.net/p/gnuplot/gnuplot-main gnuplot

開発版に対する質問は必ず mailto://gnuplot-beta@lists.sourceforge.netに送ってください。

2.4 gnuplotは私のシステムではどうコンパイルすれば良いですか

リリースノートと README, INSTALL のファイルを読んでください。C と C++ コンパイラが必要で、
様々なサポートライブラリがあらかじめインストールされている必要もありますが、それはどのような
設定オプションを選択するかによりますし、作成する実行ファイルにどの出力形式をサポートさせたい
かにもよります。

• linux 上でリリース版からコンパイルするには、./configure (一人の一般ユーザ用にインストール
する場合は、./configure --prefix=$HOME/usr)とし、次に makeとして、最後に make installです。
configureスクリプトの出力には十分注意を払ってください (そう、それはとても長いのだけれど)。
そこには、どのサポートライブラリが必要なのか、どの追加オプションが望ましいのか、といった
ことに対するヒントが現れています。一般には、各サポートライブラリの「開発版」パッケージを
インストールしておく必要があるでしょう。

• linux上で開発版ソースからコンパイルするには、./configureを実行させる前に、別のスクリプト
./prepareを実行する必要があります。

• Windows では、makefile は config/mingw, config/msvc, config/watcom, config/cygwin 内にあります。
makefileの先頭に書かれているオプションを修正し、そのmakefileが置かれているのと同じディレ
クトリ内で適当な makeツールを実行してください。追加説明も、makefileに書かれています。

• 他のプラットフォームでは、適切な makefile (例えば OS/2なら makefile.os2)を config/ ディレクト
リから src/ にコピーして、makefileのヘッダにあるオプションを必要なら更新して、srcにディレ
クトリを移動して、makeを実行してください。

2.5 ドキュメントはどこですか、それはどうすれば手に入れられますか

ソース配布物に完全なドキュメントが PDFファイル形式で含まれています。個々のセクションは、gnuplot
の作業中に helpキーワードを打つことで見ることができます。ドキュメントの他の形式も、docsサブ
ディレクトリ内でソースからコンパイルできます。

http://gnuplot.sourceforge.net/documentation.htmlに、英語版と日本語版のコピーが置いてあります。

2.6 動作するサンプル

ソース配布物には動作するサンプルのディレクトリが含まれています。これらサンプル、およびその結
果の描画グラフは http://gnuplot.sourceforge.net/demo/でも見ることができます。

2.7 私の gnuplotにどのオプションがコンパイルされているかを知るには

コンパイル済みの gnuplotが与えられている場合、showコマンドで、それがコンパイルされたときに使
用されたコンフィグレーションとビルドオプションの一覧を表示できます。あなたの gnuplotに組み込
まれている出力形式 ("terminal"とも言います)は、set terminalで表示できます。

gnuplot> show version long
gnuplot> set terminal

3 動作

3.1 ヘルプはどうすれば手に入れられますか

最初のプロンプトで helpコマンドを実行してください。その後キーワードをたどっていってください。
help plotや help setから始めるのがいいでしょう。

• もし、持っているならマニュアルを読んでください

• demoサブディレクトリのデモを実行するか、ネット上にあるそのコピーを見てください。ヒント
を与えてくれるでしょう。http://gnuplot.info/demo

• システム管理者、または gnuplotを設定した同僚に尋ねてください。

• gnuplotに関する質問を議論する Usenetグループ news://comp.graphics.apps.gnuplotが古くから
ありますが、そこの住民はほとんど前世紀の人々です。

• よりブラウザ指向のヘルプフォーラムが StackOverflowにあります。http://stackoverflow.com/questions/

tagged/gnuplot

• それらにすべて失敗した場合、質問をニュースグループ news://comp.graphics.apps.gnuplotに投
書するか、またはメーリングリスト mailto://gnuplot-info@lists.sourceforge.netにメールを送ってく
ださい。そのメーリングリストが圧倒的な量の spamを別に受けてしまうのを避けるため、投稿の
前にそのメーリングリストに参加 (subscribe)しなければいけないことに注意してください。参加
の方法は、http://lists.sourceforge.net/lists/listinfo/gnuplot-infoにあります。

質問をする際は、あなたが使用している gnuplotのバージョンとオペレーティングシステムについても
言及する必要があります。あなたのやり方で期待したようにはうまくできなかったグラフについて質問
したいときは、その問題を持つグラフを生成するような最小の gnuplotコマンド列を示すようにしてく
ださい。

3.2 どうすればグラフをプリントアウトできますか

plotコマンドが生成するものの出力形式は、その前の set terminalによって決定します。対話型でない出
力に対しては、ファイル名を指定するためにこれと set outputコマンドを組にして実行します。

以下の例の作業は、まず sin(x)のグラフをスクリーン上に描画し、同じグラフを PDF出力ファイルとし
て再描画します。注意: PDF出力は、スクリーン上に表示されるグラフと完全には同じにならない可能
性があります。

gnuplot> plot sin(x)
gnuplot> set terminal pdf
Terminal type is now ’pdfcairo’
Options are ’ transparent enhanced fontscale 0.5 size 5.00in, 3.00in ’
gnuplot> set output "sin.pdf"
gnuplot> replot
gnuplot> unset output # 出力ファイルを閉じる (でないと開いたまま)
gnuplot> unset terminal # デフォルト対話型出力形式に戻す
gnuplot>

出発点が、デフォルト対話型出力形式でない場合は、同じことを pushと popで達成できます。

gnuplot> set terminal push # 現在の出力形式を保存 (デフォルでなくてもよい)
gnuplot> set terminal pdf
gnuplot> set out ’a.pdf’
gnuplot> replot
gnuplot> unset out
gnuplot> set term pop # 保存した出力形式を復帰

対話型出力形式の中にはそのツールバー上にプリンタアイコンを提供するものもあります (win, wxt, qt
等)。それは、他の出力形式を使用するのではなく、システムに用意されている一般的なシステムツール
を使い、現在のグラフを印刷するか、またはファイルに保存します。すなわち、印刷メニューで "save to
png"を選択して得られるファイルは、set term png; replot;によって得られるファイルとは違うものにな
ります。一般に、この方法で保存したグラフは、コマンドラインから出力形式を変更して作成したもの
よりも、よりスクリーン画像に近いものを生成するでしょう。

3.3 どうすればグラフを私のワープロソフトに取り込めますか

基本的に、グラフをそのワープロソフトが理解できる形式のファイルに保存すれば、そのワープロソフト
で読み込むことができるでしょう。後で正しいサイズに変換もできるという点で、ベクトル形式のフォー
マット (PostScript, emf, svg, pdf, TEX, LATEX)の方がいいでしょう。

使用できるファイル形式の一覧を見るには set termとしてください。

多くのワープロソフトは Encapsulated PostScript (*.eps)のグラフを取り込めます。gnuplotでは、
set terminal postscript epsまたは set terminal epscairoを使うことで eps出力を選択できます。gnuplotは、
出力する EPSファイルにプレビュー用のビットマップ画像を埋め込みませんので、ワープロソフトに対
応させるために、そのファイルをワープロソフトに取り込む前に、外部ツールを使って自分自身でプレ
ビュー画像を追加しないといけない場合もあるかもしれません。

LibreOfficeやMicrosoft Office等のアプリケーションには、EMF形式のベクトル画像を処理できるもの
があります。そのような画像ファイルは emf出力形式でも作れますし、windows出力形式の描画ウィン
ドウのツールバーから ’Save as EMF...’ を選択することでも作れます。

LibreOfficeは SVG形式も、AutoCADの dxf形式も読みこむことができます。

gnuplotを使って TEXや LATEX文書に取り込むためのグラフを作る方法は何通りもあります。取り込む
*.texファイルの断片を直接生成する出力形式もありますし、\includegraphicsコマンドを使って取り込む
ための *.eps, *.pdf, *.png出力を生成するものもあります。epslatexと cairolatex出力形式は、画像ファイ
ル (*.epsか *.pdf)とそれを参照する *.tex文書ファイルの両方を生成します。tikz出力形式は、テキスト
とグラフを生成し、その出力を pdflatexで処理すれば PDFファイルが作成できます。

大抵のワープロソフトはビットマップ画像 (png, pbm等)を取り込むことができますが、この方法の短所
は、そのグラフが gnuplotによって生成されたときのサイズによってその解像度が制限されてしまうこ
とです。一般に、文書が最終的にプリントアウトされる解像度よりもその解像度はかなり低くなります。

3.4 どうすれば gnuplotのグラフを編集や後処理できますか

これはあなたが使う出力形式に依存します。

• svg 出力形式 (scalable vector graphics) の出力は Inkscape (http://www.inkscape.org), Skencil (http:

//www.skencil.org), Dia (http://projects.gnome.org/dia/),のような svgエディタで更なる編集も行なえ
ますし、OpenOffice.org (OO.o Drawプリミティブへの変換によって)の中に取り込むこともでき
ます。

• PostScript や PDF 出力は、Adobe Illustrator や Acrobat のようなツールで直接編集できますし、
pstoeditパッケージで他の色々な編集可能なベクトル形式に変換することも可能です。pstoeditは
http://www.pstoedit.netにあります。

• DXF出力形式は AutoCADの書式で、他のいくつかのアプリケーションでも編集できます。

• ビットマップ形式の画像 (png, jpeg, pbm等)は ImageMagickや Gimpの様なツールで編集するこ
とができます。一般に、ベクトルグラフィック形式の後処理を行なうにはベクトルグラフィックプ
ログラム (ドロー系ツール) を、ピクセル画像にはピクセル画像系のプログラム (ペイント系ツー
ル)を使う必要があります。

3.5 どうすれば現在の設定を保存/再利用できますか

それには save "filename"と load "filename"のコマンドを使ってください。

3.6 一つのファイルに gnuplotのコマンドとデータの両方を書けますか

gnuplotのバージョン 5では、「ヒアドキュメント」形式での名前付きデータブロックをサポートしてい
ます:

gnuplot> $DATABLOCK << EOD
cats 4 2
dogs 1 4

EOD
gnuplot> plot $DATABLOCK using 2:3:1 with labels

名前付きデータブロックは一度定義すれば、それを何度でも使うことができます。

データは、plotコマンド上で疑似ファイル指定 plot "-"を行ってインライン形式で与えることもでき
ます。この場合は、データは一度しか使えません。

gnuplot> plot "-"
1 1
2 4
3 9
e

3.7 どうすれば描画前にデータにフィルタをかけられますか

あなたの使用する OSが popen()関数をサポートしているなら、他のプログラムやシステムユーティリ
ティを plotコマンドの一部分として使用して入力データをフィルタリングすることができます。

gnuplot> plot "< sort +2 file.in" # 2 列目のデータで事前にソート

この仕組みは、Unix由来のコマンドラインツール awk, sort, grepなどと組み合わせて使用するととても
強力です。

3.8 私のプログラムで gnuplotのルーチンを利用できますか

パイプをサポートしているシステムでは、他のプログラムからパイプ経由で gnuplotにコマンドを渡す
ことができます。Octave (http://www.octave.org)のように、gnuplotをグラフィックエンジンとして使用し
ている多くのアプリケーションがこの方法を使用しています。これは、formベースのWebページから
gnuplotを起動する cgiスクリプトでも利用できます。

4 グラフの見た目のカスタマイズ

4.1 色や線や点記号のデフォルト属性を調べたり変更するには

複数の要素に対する plotや splotコマンドを実行した場合、gnuplotは、デフォルトでは色と線種の集合
を巡回的に使用します。これは、その plotコマンドに特定の色や線種属性を指定することで変更できま
すし、デフォルトの列を変更することもできます。以下のコマンドは、それぞれ追加パラメータをたく
さん指定できます。

testは、現在の出力形式に対する有効な色、線、点の属性などを表示します。

set colorと set monochromeは、定義済みの列を選択します。

set linetypeは、存在する線種の属性を変更するか、新しいものを追加します。

set paletteは、温度地図のような pm3dモードで使用されるカラーパレットを変更します。

set pointsizeは、追加の倍率としてすべての点のサイズをスケールします。

4.2 陰線処理、隠面処理

2つの関連するコマンドがあります。set hidden3dは、3次元描画スタイル splot ... with linesで描画する
曲面に影響を与えます。それは、他の 3次元描画スタイルで生成され、それらの曲面で隠される線分を
クリッピングします。しかし、pm3dモードで生成されるグラフは処理しません。with pm3d, with zerror,
with boxes のスタイル、および set pm3d が有効な最中に描かれた様々な描画要素についても同様です。
これらのグラフに対する隠面処理は、見る側からの距離の順番でそれらを描画することによって代用的
に達成されていますが、その順番は、set pm3d depthorderで制御できます。

4.3 グラフの境界をページの特定の位置に丁度合わせるには

上、下、左、右の境界位置を、ページ内の位置として比率で指定してください:

set lmargin at screen 0.05
set bmargin at screen 0.05
set rmargin at screen 0.95
set tmargin at screen 0.95

4.4 複数のグラフをひとつのページ内で隣同士に配置するには

あなたが必要とするコマンドは set multiplotです。gnuplotは指定した数のグラフを規則的な升目状に配
置します (set mulplot layout <rows>, <columns> ...)。または、それらを set originと set sizeを使ってひ
とつずつ位置を決めることもできます。

4.5 縦と横の軸の目盛りを 1:1にできますか

set size squareか set view equal xyとしてみてください。

4.6 カラーとモノクロの両方の出力で機能するようなパレットは

set palette cubehelixを試してください。

4.7 どうしたらデータ点をスキップできますか

以下のようにデータ値として ? を指定すればできます。

1 2
2 3
3 ?
4 5

set missingも参照してください。データファイルのコメント文字を指定するには、set datafile commentschars
を参照してください。

4.8 どうしたら n点毎にデータを扱えますか

これは、例えば plot ’a.dat’ every 2のように、plotコマンドに色々なオプションを指定することでできま
す。すべての点を通る線を引き、しかし点の記号は n点毎に書きたい場合は、
plot ’a.dat’ with linespoints pointinterval nとしてみてください。

4.9 どうしたら鉛直線を書けますか

状況によりますが、主な方法は以下の通りです:

• set arrow .... .... nohead この場合、矢の正確な開始点と終了点の位置を計算しておく必要があり
ます。

• データ点を作って (インラインでの指定も可)それを描画

5 良く質問されるグラフの形式

5.1 アニメーション

gnuplot出力形式で直接アニメーションファイルを出力できるのは以下のもの (gif)だけです:

set terminal gif animate {delay <time>} {loop <N>} {optimize}

デモの中の http://gnuplot.sourceforge.net/demo/animate.htmlも参照してください。

5.2 陰関数のグラフ

gnuplotでは陰関数のグラフや曲線は直接は描画できません。しかし、次善の策はあります。

gnuplot> # 例: 次の行を適当な定義に置き換えてください。
gnuplot> f(x,y) = y - x**2 / tan(y)
gnuplot> set contour base
gnuplot> set cntrparam levels discrete 0.0
gnuplot> unset surface
gnuplot> set table $TEMP
gnuplot> splot f(x,y)
gnuplot> unset table
gnuplot> plot $TEMP w l

その仕掛けは、曲面 z=f(x,y)の 1本の等高線 z=0を描き、その等高線のデータを一時ファイルやデータ
ブロックにセーブすることにあります。

5.3 2つの関数の間の領域を塗りつぶすには

2つの関数 f(x)と g(x)の間を塗りつぶすグラフは、疑似ファイル ’+’と filledcurvesを使えば簡単にでき
ます。

f(x)=cos(x); g(x)=sin(x)
set xrange [0:pi]
plot ’+’ using 1:(f($1)):(g($1)) with filledcurves

注意してもらいたいのですが、上の例のスクリプトは 2つの曲線の間の領域を塗りつぶすものであって、
どちらが他方の上であるかを考慮はしません。もし、g(x)<f(x)か、または f(x)<g(x)を満たす領域のみ
を塗り潰したいなら、filledcurvesの後ろに aboveか belowのキーワードを追加してください。

5.4 3次元データの 2次元射影を描画するには

コマンド set view mapは、視角とスケールを調整し、それに続く splotによる 3次元グラフを、plotによ
る 2次元グラフとほぼ同じレイアウトで表示してくれます。

5.5 pm3d色地図/曲面上に dots/pointsの散布図を重ねるには

pm3dスタイルの explicitオプションを使ってください:

gnuplot> set pm3d explicit
gnuplot> splot x with pm3d, x*y with points

5.6 ラベルつきの等高線を描画するには

個々の等高線にラベルをつけたグラフは、gnuplotバージョン 4ではある種の裏技と余分な処理手順が
必要でしたが、バージョン 5ではその手順はとても簡単です。等高線を、1回目は "with lines"で、2回
目は "with labels"で 2回描画するだけです。目立つラベルを作成したい場合は、以下のようにするとい
いでしょう。

set style textbox opaque noborder

set contours

splot ’DATA’ with lines, ’DATA’ with labels boxed

13



5.7 gnuplotは棒グラフ/ヒストグラム/箱型グラフが書けますか

gnuplot は、あらかじめ表形式にされたデータを表示するための、さまざまな横並び型 (clustered)と積
み上げ型 (stacked)のヒストグラムスタイルをサポートしています。また、生のデータを場所毎に累積す
るためのいくつかのオプションも提供していて、それを切り替えて棒グラフとして表示させることも可
能です。マニュアルの binsと smooth frequencyの部分を参照してください。

5.8 gnuplotは円グラフが書けますか、四半期グラフはどうですか

円グラフは gnuplotではちょっと面倒ですが、以下を見てください。http://gnuplot.sourceforge.net/demo/

circles.htmlまたは、以下を参照してください。http://gnuplot-tricks.blogspot.com/2009/08/pie-charts-entirely-

in-gnuplot.html

デモの中には単純なガントチャートの例があります。

5.9 gnuplotは一つのグラフで複数の y軸を使えますか

はい。2次元グラフでは別々に x軸を下 (x1)と上 (x2)に持つことができますし、別々に y軸を左 (y1)
と右 (y2)に持つことができます。バージョン 5では、with parallelaxesという描画モードで y軸をさら
に追加することもできます。

5.10 小 3角面で構成される物体を描画するには

3次元物体の面に色をつけたものをファイルに記述するには、以下のような形式にすれば可能です。

# 三角形 1
x0 y0 z0 <c0>
x1 y1 z1 <c1>

x2 y2 z2 <c2>
x2 y2 z2 <c2>

# 三角形 2
x y z
...

1行の空行と 2行の空行に注意してください。また、三角形は実際には、4つの頂点を記述し、そのう
ちの 2つを同一とすることにも注意してください。<cN>はオプションの色指定です。

これを以下のようにして描画します (どちらかの splotで):

set pm3d
set style data pm3d
set pm3d depthorder
splot ’facets.dat’
splot ’facets_with_color.dat’ using 1:2:3:4

gnuplotは 3次元モデリングソフトではありませんので、depthorderによる描画では、面同士の貫通は処
理されません。本当の 3次元描画のためには、多分レイトレーシングソフトを使う方がいいでしょう。

5.11 どうしたら 2つの関数を重ならないように書けますか

描画する各関数の直前に描かせたい範囲を置いてください。例えば、実験データと、領域の別の部分を
カバーする 2つの異なる関数モデル f1, f2を描画するには:

gnuplot> set autoscale x # x の範囲をデータから取得
gnuplot> plot ’data’, [*:0] f1(x), [0:*] f2(x)

5.12 どうしたら splotで格子でない曲線を書けますか

splotへの入力データが、各行が同じデータ点数 (通常は空行で分離されている)を持つように揃っていれ
ば、splotはデフォルトではそのデータを曲面を記述するものとして扱います。そうでなく個々の曲線を
描画したければ、unset surface, set surface explicit, plot ... nosurfaceなどを組み合わせてみてください。

5.13 どうしたら x-y平面上で他の関数に制限されている f(x,y)のグラフを書けますか

以下は一つの方法です:

gnuplot> f(x,y) = x**2 + y **2
gnuplot> x(u) = 3*u
gnuplot> yu(x) = x**2
gnuplot> yl(x) = -x**2
gnuplot> set parametric
gnuplot> set cont
gnuplot> splot [0:1] [0:1] u,yl(x(u))+(yu(x(u)) - yl(x(u)))*v,\

> f(x(u), (yu(x(u)) - yl(x(u)))*v)

6 文字列の整形と特別な記号

6.1 "enhanced text"モードによる文字列のマークアップ

gnuplotバージョン 5では拡張文字列モード (enhanced text)がデフォルトで、それは特殊文字を文字列
内に埋めこむことによって文字列のマークアップを示します。

拡張文字処理制御記号
制御記号 例 結果 説明

^ a^x ax 上付き文字
_ a_x ax 下付き文字
@ a@^b_{cd} ab

cd 空ボックス (幅がない)
& d&{space}b d b 指定した長さのスペースを挿入
~ ~a{.8-} ã ’a’の上に ’-’を、現在のフォントサ

イズの .8倍持ち上げた位置に重ね書き
{/Times abc} abc Timesフォント、今のサイズで abcを出力
{/Times*2 abc} abc Timesフォント、今の倍のサイズで abc
{/Times:Italic abc} abc Timesフォント、イタリック体で abc
{/Arial:Bold=20 abc} abc Arialフォント、太字、サイズ 20で abc

6.2 文字列のマークアップをオンに/オフにするには

この処理モードから、特定のテキスト文字列を除外するには、キーワード noenhancedを使用してくだ
さい。例えば、ファイル名を下付き文字と解釈させないためには:

set title ’Compare file_1.dat and file_2.dat’ noenhanced

6.3 特殊文字に関するすべて問題の回答は UTF-8にすることですか

ええ。

不運にも、特に PostScript出力の生成時など、使用するにはかなり厄介な環境がいくつかあります。UTF-8
のコンピュータ環境で作業をしている場合、gnuplotでそれを使用する場合には、多分何か特別なことを
する必要はないでしょう。しかしそうでない場合は、出力する際には gnuplotに UTF-8を使用すること
を以下のように伝える必要があります: set encoding utf8

UTF-8文字列をあなたのキーボードで入力できない場合は、gnuplotの外でそれを解決するか、または
その 8進のエスケープ文字列を 1バイト毎入力してください。あなたのキーボードで UTF-8を生成でき
るがどのキーストロークで特定の文字列を生成できるかを知らない場合は、多分適当な文字選択アプリ
ケーションがあなたのデスクトップについているでしょう (例えば KDEなら kcharselect、GNOMEなら
Character Map)。

6.4 Planck定数 (hに横線)が必要なときはどうしたらいいですか

最も自然な方法は、UTF-8フォントを使い、文字 h̄ (Unicodeコードポイント #x210F)を直接入力するこ
とです。

PostScript: PostScriptでは utf8の処理は面倒なので、拡張文字列処理 (enhanced textモード)のマークアッ
プと、多分特別な Symbolフォントを使った擬似的なことをやる必要があります:

@{/=56 -} {/=24 h} または {/=8 @{/Symbol=24 -} _{/=14 h}} 後者は 24pt (ポイント)
の "-" (/Symbolフォントの中の長いハイフン)をスペースなしで使います。14ptの "h"は 8ptのスペース
("_"の前にあるスペース)だけずらして置きますが、それは下付き文字として書くので小さい文字にな
ります。しかしそれらは通常我々が使用するプランク定数の記号とは違っていて、横線は斜線ではなく
水平線になります。もう一つの方法は {/=14 @^{/Symbol=10 -}{/=14 h}}とすることです。

AMS-LaTeX の PostScript フォント (http://www.ams.org/tex/amsfonts.html にあり、そして多くの LaTeX
配布にも含まれています) を使えば、とても簡単に小さい Planck 定数をセットすることができます
が、gnuplot (help fontpath を参照) と、PostScript インタープリタ (通常 Ghostscript) がファイル
msbm10.pfb (または msbm10.pfa)がどこにあるかを知る必要があります。"h"に傾斜のついた横棒
を重ねた \hslashを生成するには {/MSBM10 \175}を使ってください。通常の \hbar (水平の
横棒のもの)は 8進数の 176です。ただし、水平の横棒はイタリック体としてしか存在しないことに注
意してください。

6.5 太陽質量記号 (丸に点)が必要なときはどうしたらいいですか

Planck定数と同様、最も自然な方法は UTF-8フォントを使い、文字 ⊙ (Unicodeコードポイント #x2299
; "circled dot operator")を直接入力することです。非常によく似たグリフがコードポイント #x2609にも
あります。これは "sun"で、よりいいかもしれませんが、それを提供するフォントは多くはありません。

6.6 ギリシャ文字や他の特別な記号を使うには

古いやり方として、特別なフォント、例えば Adobeの "Symbol"フォントに切り替えるために拡張文字
処理モード (enhanced textモード)を使用する方法があります。それは、あなたが出力したい文字に通常
の ASCII文字を割り当てています。これは、現在でも PostScript出力には必要でしょう。しかし、より
簡単な方法は、UTF-8エンコーディングを選択し、特殊文字を他の文字列と同様にそのまま入力するこ
とです。この方法では、フォントの変更は必要なく、また CJK文字集合を含むすべての Unicodeコード
ポイントへのアクセスが可能になります。しかし gnuplotによって作られたファイルを実際に印刷、表
示するには、あなたのコンピュータ、あるいは出力デバイスに適切なフォントがインストールされてい
る必要があります。gnuplot自身にはフォントは含まれていません。

いくつかの LATEX系出力形式 (latex, epslatex, tikz, context, cairolatex)は、生成した文字列を LATEXに渡し
ます。この場合は、’\alpha_{3}’のような、通常の LATEX風の記述が利用できます。

6.7 どうしたらアクセント付きの文字を出力できますか

utf8ではないエンコーディングを使い続けるのであれば、ü (uウムラウト; uの上に点 2つのアクセント)
や n̂のような文字をラベルに含めるには、適切なエンコードオプションを設定し、8ビット文字コード
を使用しなければいけません。エスケープシーケンスを使って 8ビットコードを表現することも可能で
す。例えば次のようにします。

gnuplot> set encoding iso_8859_1
gnuplot> set title "M\374nchner Bierverbrauch \374ber die Jahre"
gnuplot> plot "bier.dat" u 1:2

しかし、そうでない人は UTF-8を使うべきで、そこではそれらは普通の文字になります。

6.8 異なるサイズのテキストを一つのグラフで出力できますか

たいていの出力形式では、デフォルトフォントの名前とサイズを指定できます。「拡張文字処理」モード
("enhanced text")では、フォント、および文字サイズやボールド、イタリックの書体を一つのグラフ内で
変更できます。

7 その他

7.1 gnuplotには画像形式 XXXをサポートするドライバはありますか

インストール済の gnuplotが使用できるグラフィックドライバの一覧を見るには、set termとタイプして
ください。

通常の配布物には含まれていても、デフォルトでは使えないようになっているグラフィックドライバも
あります。それらが使いたければ、ソースからコンパイルし直す必要があります。

7.2 バグを見つけたのですがどうしたら良いですか

最初に、それが本当にバグなのか、それともある set–commandで翻すことができる特徴なのかを見極め
てください。

次に、あなたが使っている gnuplotが古い版でないか確認してください。もしそうなら、新しい版でそ
のバグは取り除かれている可能性があります。

開発版は、現在のリリース版以後に報告されたバグの修正を既に持っている可能性があります。レポー
トを送る前に、質問のバグが既に修正されていないかチェックしてみてください。

これらをチェックした後でなおそれがバグであると確信したなら、http://sourceforge.net/p/gnuplot/bugsの
bug-trackerで報告してください。その際、gnuplotのバージョン (パッチレベルも含む)とそれが動いて
いるオペレーティングシステムも書いてください。特にそのエラーを再現する簡単なスクリプトを提供
してもらえると大いに助かります。

souceforgeの trackerは、バグのレポート、および後のリリースに反映するためのバグの修正の収集のた
めのものです。gnuplotに関する問題に対する回避策を見つける、あるいはそれを実際に解決するには、
インターネット上に色々存在するフォーラムの方がよい場所でしょう。

7.3 どう追加や拡張がなされていますか、それはどう手に入れられますか

一般的に、現在の開発版は、最終安定リリース版の gnuplotにはまだ入っていない機能も入っています。
2019年 4月現在、開発版のバージョン 5.3は、以下を特にサポートしています:

• 64ビット整数計算
• 3次元ボクセル格子 (voxel grid)
• 3次元描画スタイル isosurface, circles, boxes
• コマンド set wallsで 3次元グラフに xy, xz, yz平面を表示
• Unicodeエスケープシーケンス \U+hhhh
• 第 1種、第 2種の N次ベッセル関数
• 変形ベッセル関数
• 複数の textboxスタイル

7.4 "heavy-duty"データを gnuplotで使えますか、または gnuplotに代わるものはないですか

gnuplotそれ自身は、しんどい数値計算にはあまり向いていません。しかし、かなり大きなデータ集合
(100万データ点程度)は簡単に処理できます。簡単ではない関数の評価のための gnuplotの速度が制限を
与えてしまうような特定のアプリケーションを使用したい場合は、その関数を Cか C++でコーディン
グして作り、それをプラグインとして gnuplotで呼び出す、という方法もあります。

さらなる数値計算を必要とする場合は、以下の 2つのソフトを調べてみてください:

octave (http://www.octave.org)は、高水準言語で、主に数値計算用です。Octaveは GPLライセンスに従
い、原則的にフリーな Matlab クローンです。線形や非線形の問題を数値的に解くのに便利なコマンド
ラインインターフェースを備えています。なお、octaveは gnuplotを描画エンジンとして使いますので、
gnuplotの上位のデータ処理ソフトを手に入れることにもなります。

scilab (http://www.scilab.org)は、別の、matlabに代わるオープンソースソフトです。

7.5 gnuplotが持たない特殊関数が必要な場合は

上を見てください。他のソフトを使うか、カスタムプラグインを書くか、のいずれかが必要です。

7.6 私の対話型ターミナルでホットキーを使うには

ほとんどの対話型出力形式では、再描画、各要素の描画のトグル、軸のスケールの変更などを行うため
に、あらかじめ設定されたホットキー、およびユーザ定義可能なホットキーの両方をサポートしていま
す。現在有効な gnuplotの描画ウインドウ上で hを打つと、ホットキーの一覧が表示されます。さらな
る情報については、help mouseと help bindを参照してください。

7.7 gnuplotの次期バージョンの開発について知りたいのですがどうしたらいいですか

gnuplotのベータテストメーリングリストに入ってください。メール本文 (メールタイトルではなく)に
subscribe gnuplot-betaと書いてメールを mailto://Majordomo@lists.sourceforge.netに送ってくだ
さい。

8 共通の問題

8.1 助けて! 私のフォントが使えない

gnuplotは、それ自身ではフォント処理を行いませんので、それは必然的に個々のデバイスをサポートす
るライブラリに任せることになります。残念ながらこれは、フォントの検出においては、異なる出力形
式には異なる説明が必要だということを意味しますので、ここでは、簡単なヒントをいくつか上げてお
きます。詳しい情報については、あなたが問題にしている出力形式用の gnuplotのドキュメントを参照
してください。

png/jpeg/gif これらの出力形式は、libgdというサポートライブラリを使用し、これは環境変数 GDFONT-
PATHで与えられるディレクトリ内でフォントを探します。そのように libgdのフォントパスを設
定すると、今度は gnuplotのデフォルトフォントを設定したいかもしれません。例:
setenv GNUPLOT_DEFAULT_GDFONT verdana

post PostScriptフォント名は、その文書が印刷されるまで参照されませんから、あなたのプリンタでど
のフォントが有効であるかを gnuplot は知りません。よって、どんなフォント名でも指定するこ
とができます。しかし、gnuplot の出力にフォントを埋め込むことは可能です。それについては、
gnuplotの内部コマンド “help set term post fontfile”で得られる指示を参照してください。

svg フォント処理は、ビューワ依存です。

x11 x11出力形式は、通常の x11フォントサーバ機構を使用しますが、少しだけトリッキーな点があり、
マルチバイトフォントを使用するためには、明示的に以下のように指定する必要があります:

set term x11 font "mbfont:sazanami mincho,vera,20"

win ツールバーの "Options"のプルダウンメニューから "Choose font..."を選択します。

wxt, qt Linuxシステムでは、これらの出力形式はシステムの fontconfigユーティリティが提供するフォ
ントを使用します。

8.2 qt出力形式のセッションでの最初のグラフが失敗したり変なレイアウトになります

"slow font initialization"というエラーメッセージも出たかもしれません。これは、qtが共有システムフォ
ントキャッシュに依存しているからです。もし、他の誰も使っていないフォントを要求すると、そのキャッ
シュの更新に時間がかかります。これは、Windowsや OSXではよく起こるのですが、それらのシステム
では他のソフトが異なるフォントの仕組みを使用する傾向にあるため、関連するフォントキャッシュが
空になるからです。gnuplotの呼び出し時に、コマンドラインオプション --slowをつけてみてください。

8.3 pm3dによるデータファイルの splotで何も表示されません

set pm3d; splot ’a.dat’ とやっても、カラーボックスは表示されるもののグラフは表示されないとした
ら、もしかしたら 2つのつながっている線 (孤立線)を分離する空行がないのでは ? 空行を入れなさい
! これが何を意味するのかに興味があるなら、gnuplot の demo ディレクトリにある demo/glass.dat や
demo/triangle.datなどのファイルをじっくり見てください。

以下の awkスクリプト (例えば addblanks.awkと呼ぶことにします)は有用で、データファイルに、第 1
列目の数が変更したときに空行を追加してくれます。

/^[[:blank:]]*#/ {next} # コメント行 (# で始まる行) は無視
NF < 3 {next} # 3 列未満しかない行も無視
$1 != prev {printf "\n"; prev=$1} # 空行を出力
{print} # その行自体を出力

これにより、データファイルをコマンド awk -f addblanks.awk <a.datによって前処理するか、unix互換
のプラットフォーム上でデータファイルを
gnuplot> splot "<awk -f addblanks.awk a.dat"のように描画するか、のどちらかを行ってください。

8.4 なぜ gnuplotはとても小さい数字のデータを無視するのですか

gnuplotは、ある目的のため 1e-08より小さい数を 0として扱います。よってとても小さい数の集まり
を描画しようとするとそれらは 0として描画されてしまいます。対数軸で描画する場合は、もっとひど
いことにそれらは目盛りからはずれてしまいます。また、全ての数が "0"である場合、範囲は空である
とみなされます:

gnuplot> plot ’test1’
Warning: empty y range [4.047e-19:3e-11], adjusting to [-1:1]
gnuplot> set yrange [4e-19:3e-11]
gnuplot> plot ’test1’

^
y range is less than ‘zero‘

その対策は gnuplotの "zero"の概念を変更することです。

gnuplot> set zero 1e-20

より詳しい情報については help set zeroを参照してください。

8.5 multiplotのグラフを replotすると一かけらしか出ません

gnuplotは、直近の plotや splotコマンドを再生成するために十分な情報を保持するだけなので、multiplot
に対しては、それを生成する完全なコマンド列をスクリプトファイルに保存する必要があります。そし
て、loadすることでその内容を gnuplotに取り込めば、あなたが必要とする回数だけ異なる出力形式、
異なる出力ファイルに対して再描画をすることができます。

8.6 書いた式 (1/3など)が意味のない結果を出力してしまいます

gnuplotは整数の式の場合、実数演算ではなく整数演算を行ないます。例えば式 1/3は 0と評価されま
す。もし実数の値が欲しいならば、その数の後にドット "."をつけてください。例:

gnuplot> print 1/3
0

gnuplot> print 1./3.
0.333333

整数の式をこのようにして評価するやり方は Cや Fortranでも行なわれています。

8.7 私の出力が不完全です

set outputか unset outputによって、たまっている出力を吐き出させ、そして出力をクローズする必要が
あるかも知れません。

出力形式によっては、一つの出力ファイルに複数のページが含まれています (postscript, pdf, latex, svg,
...)。これらの出力モードでは、gnuplotはあなたが描画の追加ができるように各描画後にファイルをオー
プンしたままにしています。よって異なる出力形式を選択する (set term) か、または gnuplot を終了す
るまでは、そのファイルは完全ではなく、外部のアプリケーションでは使用できるようにはなっていま
せん。

8.8 パイプやスクリプトから gnuplotを呼び出すとスクリーンにグラフがでません

よくある原因の一つが、gnuplotがグラフの描画直後に終了してしまうことで、終了時に描画ウィンド
ウが閉じてしまうため、あなたがグラフを見れないぐらいにすべての表示があっという間に終わってし
まう、という状態です。これには、いくつかの解決法があります。

• gnuplotのコマンドラインオプション -persistを使用すること。これは、gnuplot自身の終了後も描
画ウィンドウを残します。

• オプション set termで persistを受け付ける対話型出力形式を使用すること。

• パイプの終了前に gnuplotにコマンド pause mouse closeを送信すること。これは、描画ウィンド
ウが他の動作で終了するまで gnuplotを終了させないように指示します。

• あなたが gnuplot、および描画ウィンドウ等々を本当に必要としなくなった場合にのみ、スクリプ
トにパイプを閉じさせること。

以下は、これらの修正のうち 2つを示す Perlスクリプトです:

21



#!/usr/local/bin/perl -w
open (GP, "|/usr/local/bin/gnuplot -persist") or die "no gnuplot";
# 各出力後に強制的にバッファを掃き出す
use FileHandle;
GP->autoflush(1);
print GP,"set term x11; plot sin(x) with lines\n";
print GP,"pause mouse close\n";
close GP

9 帰属

この FAQは、Russell Lang, John Campbell, David Kotz, Rob Cunningham, Daniel Lewart, Alex Wooの寄与
を John Fletcherが最初にまとめたものです。そして、Alex Wooの草案により、Alex Woo, John Campbell,
Russell Lang, David Kotzらの修正と追加、Daniel Lewartによる多くの修正が Thomas Koenigによりまと
め直されました。そして、gnuplot 3.7用に再び Lars Hecking, Hans-Bernhard Broekerおよびその他の人々
による修正が Alexander Maiと Juergen v.Hagenによりまとめ直されました。gnuplotバージョン 4用の
修正は、Petr Mikulíkと Ethan Merrittにより行なわれました。gnuplotバージョン 5用の修正は、Ethan
Merrittにより行なわれました。

(訳注: 日本語訳は Shigeharu TAKENO (mailto://shige@iee.niit.ac.jp)により行なわれました。)

